

Extract the Envelope and Instantaneous Phase of a Signal using the Hilbert Transform

Introduction

The [Hilbert transform](#) has many practical applications. These include vibration analysis and amplitude modulation in communication systems

This application compute the envelope and instantaneous frequency of a signal using the Hilbert transform.

- The signal envelope is the magnitude of the Hilbert transform
- The instantaneous frequency is the derivative (wrt time) of the phase angle of the Hilbert transform (only for single-component signals)


The magnitude of the analytic signal captures the slowly varying features of the signal, while the phase contains the high-frequency information.

> restart:

```
plot_opts := thickness = 3, color = "DarkGreen", font = [ Arial ],
axes = boxed, size = [ 800, 400 ], background = ColorTools:-Color(
"RGB", [ 218 / 255, 223 / 255, 225 / 255 ] ), axis = [ gridlines =
[ 5, color = ColorTools:-Color( "RGB", [ 1, 1, 1 ] ) ] ]:
```


Signal

```
> t := Vector( [ seq( 0..10, 0.01 ) ], datatype = float[ 8 ] ):
signal := (1 +~ sin~( t )) *~ sin~( 10 * t ) :
p1 := plot( t, signal, plot_opts, title = "Signal" );
```



Hilbert Transform, Signal Envelope and Instantaneous Phase

```
> envelope := abs( SignalProcessing:-Hilbert( signal ) ):
p2 := plot(t, envelope, color = "DarkBlue" ):
p3 := plot(t, -envelope, color = "DarkBlue" ):
plots:-display( p1, p2, p3, plot_opts, title = "Signal Envelope
via the Hilbert Transform" )
```



```
> phase := MTM:-unwrap( Re~( argument~( convert~( SignalProcessing:-
Hilbert( signal ), float) ) ) ):
plot(t, phase, title = "Instantaneous Phase of the Signal",
plot_opts)
```

Instantaneous Phase of the Signal

